Problem:
Score : 300 points

Problem Statement

AtCoDeer the deer is seeing a quick report of election results on TV.
Two candidates are standing for the election: Takahashi and Aoki.
The report shows the ratio of the current numbers of votes the two candidates have obtained, but not the actual numbers of votes.
AtCoDeer has checked the report N times, and when he checked it for the i-th (1≦i≦N) time, the ratio was T_i:A_i.
It is known that each candidate had at least one vote when he checked the report for the first time.

Find the minimum possible total number of votes obtained by the two candidates when he checked the report for the N-th time.
It can be assumed that the number of votes obtained by each candidate never decreases.

Constraints

1≦N≦1000

1≦T_i,A_i≦1000 (1≦i≦N)

T_i and A_i (1≦i≦N) are coprime.

It is guaranteed that the correct answer is at most 10^{18}.

Input

The input is given from Standard Input in the following format:

N
T_1 A_1
T_2 A_2
:
T_N A_N

Output

Print the minimum possible total number of votes obtained by Takahashi and Aoki when AtCoDeer checked the report for the N-th time.

Sample Input 1

3
2 3
1 1
3 2

Sample Output 1

10

When the numbers of votes obtained by the two candidates change as 2,3 → 3,3 → 6,4, the total number of votes at the end is 10, which is the minimum possible number.

Sample Input 2

4
1 1
1 1
1 5
1 100

Sample Output 2

101

It is possible that neither candidate obtained a vote between the moment when he checked the report, and the moment when he checked it for the next time.

Sample Input 3

5
3 10
48 17
31 199
231 23
3 2

Sample Output 3

6930

Solution in Fortran (gfortran v4.8.4) (judge verdict: Accepted)

program main
  implicit none
  integer(8) :: n, i, x, y
  integer(8), allocatable :: a(:), t(:)
  read(*, *) n
  allocate(a(n), t(n))
  do i = 1, n
    read(*, *) t(i), a(i)
  end do
  x = 1
  y = 1
  do i = 1, n
    call update(x, y, t(i), a(i))
  end do
  write(*, "(i0)") x + y
  deallocate(a, t)

  contains

  subroutine update(x, y, t, a)
    implicit none
    integer(8), intent(inout) :: x, y
    integer(8), intent(in) :: t, a
    integer(8) :: ma, mb, mult
    ma = x / t
    if (mod(x, t) > 0) ma = ma + 1
    mb = y / a
    if (mod(y, a) > 0) mb = mb + 1
    mult = max(ma, mb)
    x = t * mult
    y = a * mult
  end subroutine update
end